برین‌فاک
پیاده سازی مفسر فاک برین در c > مثال‌ها

به عنوان مثال نخست، قطعه کد زیر مقدار سلول فعلی را به مقدار سلول بعدی اضافه می‌کند: هر بار که حلقه اجرا شود، سلول فعلی کاهش می‌یابد ، نشانگر داده به سمت راست حرکت می‌کند و سلول بعدی افزایش می‌یابد و نشانگر داده دوباره به سمت چپ حرکت می کند. این دنباله تا زمانی که سلول اولیه به صفر می‌رسد تکرار می‌شود.
[->+<]
این کد می‌تواند در یک برنامه جمع ساده مانند مثال زیر گنجانیده شود.
++ Cell c0 = 2
> +++++ Cell c1 = 5

[ Start your loops with your cell pointer on the loop counter (c1 in our case)
< + Add 1 to c0
> - Subtract 1 from c1
] End your loops with the cell pointer on the loop counter

At this point our program has added 5 to 2 leaving 7 in c0 and 0 in c1
but we cannot output this value to the terminal since it is not ASCII encoded!

To display the ASCII character "7" we must add 48 to the value 7
48 = 6 * 8 so let's use another loop to help us!

++++ ++++ c1 = 8 and this will be our loop counter again
[
< +++ +++ Add 6 to c0
> - Subtract 1 from c1
]
< . Print out c0 which has the value 55 which translates to "7"!
این برنامه عبارت Hello World را خروجی می‌دهد:
[ This program prints "Hello World!" and a newline to the screen, its
  length is 106 active command characters [it is not the shortest.]

  This loop is a "comment loop", it's a simple way of adding a comment
  to a BF program such that you don't have to worry about any command
  characters. Any ".", ",", "+", "-", "<" and ">" characters are simply
  ignored, the "[" and "]" characters just have to be balanced.
]
+++++ +++ Set Cell #0 to 8
[
  >++++               Add 4 to Cell #1; this will always set Cell #1 to 4
    [                   as the cell will be cleared by the loop
      >++             Add 2 to Cell #2
      >+++            Add 3 to Cell #3
      >+++            Add 3 to Cell #4
      >+              Add 1 to Cell #5
        <<<<-           Decrement the loop counter in Cell #1
    ]                   Loop till Cell #1 is zero; number of iterations is 4
  >+                  Add 1 to Cell #2
  >+                  Add 1 to Cell #3
  >-                  Subtract 1 from Cell #4
  >>+                 Add 1 to Cell #6
    [<]                 Move back to the first zero cell you find; this will
                        be Cell #1 which was cleared by the previous loop
    <-                  Decrement the loop Counter in Cell #0
] Loop till Cell #0 is zero; number of iterations is 8

The result of this is:
Cell No : 0 1 2 3 4 5 6
Contents: 0 0 72 104 88 32 8
Pointer : ^

>>. Cell #2 has value 72 which is 'H'
>---. Subtract 3 from Cell #3 to get 101 which is 'e'
+++++++..+++. Likewise for 'llo' from Cell #3
>>. Cell #5 is 32 for the space
<-. Subtract 1 from Cell #4 for 87 to give a 'W'
<. Cell #3 was set to 'o' from the end of 'Hello'
+++.----.----. Cell #3 for 'rl' and 'd'
>>+. Add 1 to Cell #5 gives us an exclamation point
>++. And finally a newline from Cell #6
در اینجا برای خوانا کردن این کد، فاصله‌ها و کامنت‌های زیادی اضافه شده‌اند. برین‌فاک همهٔ کاراکترهای غیر از +-<>[],. را نادیده می‌گیرد. برنامهٔ بالا به صورت خلاصه چنین می‌شود:
نوشتار IRAN به زبان برین فاک
+++++-++++[>+++++++++>><++++-+++++++>++++++++<<<-]>+.>++.>+.<----.
نوشتار Hello World!
++++++++[>++++[>++>+++>+++>+<<<<-]>+>+>->>+[<]<-]>>.>---.+++++++..+++.>>.<-.<.+++.------.--------.>>+.>++.